

Integrals, Partial Fractions, and Integration by Parts

In this worksheet, we show how to integrate using Maple, how to explicitly implement integration by parts, and how to convert a proper or improper rational fraction to an expression with partial fractions.

Integrals

As a first example, we consider $\int \frac{x}{x^3 - 1} dx$. We begin by entering $\frac{x}{x^3 - 1}$ as a Maple expression by using an [assignment](#) statement.

```
> restart;  
> f:=x/(x^3-1);
```

$$f := \frac{x}{x^3 - 1}$$

We can use the [eval](#) command to evaluate f at $x = 3$, for instance.

```
> eval(f,x=3);
```

$$\frac{3}{26}$$

We indicate the integral by using the [Int](#) command. This is the inert form of the integral.

```
> Int(x/(x^3-1),x);
```

$$\int \frac{x}{x^3 - 1} dx$$

We can use the [value](#) command, which evaluates inert functions, to evaluate the integral.

```
> value(Int(x/(x^3-1),x));
```

$$-\frac{1}{6} \ln(x^2 + x + 1) + \frac{1}{3} \sqrt{3} \arctan\left(\frac{1}{3} (2x + 1) \sqrt{3}\right) + \frac{1}{3} \ln(x - 1)$$

We can also use [int](#) to evaluate the integral. The `int` command evaluates the indefinite integral by just producing its anti-derivative (without the constant of integration), and does not print the integral symbol.

```
> int(x/(x^3-1),x);
```

$$-\frac{1}{6} \ln(x^2 + x + 1) + \frac{1}{3} \sqrt{3} \arctan\left(\frac{1}{3} (2x + 1) \sqrt{3}\right) + \frac{1}{3} \ln(x - 1)$$

For a full presentation of the integral formula, we use:

```
> Int(x/(x^3-1),x)=int(x/(x^3-1),x);
```

$$\int \frac{x}{x^3 - 1} dx = -\frac{1}{6} \ln(x^2 + x + 1) + \frac{1}{3} \sqrt{3} \arctan\left(\frac{1}{3} (2x + 1) \sqrt{3}\right) + \frac{1}{3} \ln(x - 1)$$

We next consider the function $f(x) = \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}$. We enter this into

Maple by using the Maple [functional notation](#).

```
> f:=x -> (2*x^5-8*x^4+15*x^3-10*x^2-9*x+27)/(x^4-4*x^3+5*x^2-4*x+4);
```

$$f := x \rightarrow \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}$$

This allows for easy evaluation of the function for different values.

> **f(3);**

$$\frac{153}{10}$$


We give a full presentation of this integral. Note the use of $f(x)$ as opposed to f above.

> **Int(f(x),x)=int(f(x),x);**

$$\int \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4} dx = x^2 - \frac{5}{x-2} + 3 \ln(x-2) + \ln(x^2 + 1) + 7 \arctan(x)$$

Int and int can also be used for definite integration as seen in the following. Suppose we wish to integrate our function from $x = -2$ to $x = 1$. We first plot this part of the graph to make sure the function is continuous there.

> **plot(f(x),x=-2..1);**



Since the function is continuous over our interval of integration, we may proceed. Notice how we include the limits of integration.

> **Int(f(x),x=-2..1)=int(f(x),x=-2..1);**

$$\int_{-2}^1 \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4} dx = -\ln(5) + 7 \arctan(2) - 5 \ln(2) + \frac{3}{4} + \frac{7}{4} \pi$$

We can incorporate the `evalf` command to get a numerical approximation to the exact integral.

> Int(f(x), x=-2..1)=evalf(int(f(x), x=-2..1));

$$\int_{-2}^1 \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4} dx = 8.922654355$$

Partial Fractions

We now wish to use Maple to convert the integrand above by the method of partial fractions. When doing partial fractions by hand, we use the method only with proper fractions. Here, we use the `convert` command, where the argument `parfrac` refers to partial fraction format. This will work with improper fractions also.

> pf:=convert(f(x), parfrac, x);

$$pf := 2x + \frac{2x + 7}{x^2 + 1} + \frac{3}{x - 2} + \frac{5}{(x - 2)^2}$$

> Int(pf, x)=int(pf, x);

$$\int \left(2x + \frac{2x + 7}{x^2 + 1} + \frac{3}{x - 2} + \frac{5}{(x - 2)^2} \right) dx = x^2 - \frac{5}{x - 2} + 3 \ln(x - 2) + \ln(x^2 + 1) + 7 \arctan(x)$$

This is the same answer that we got above. To illustrate further, let us create separate expressions for the numerator and denominator of $f(x)$. We will use the `numer` and `denom` commands to do this.

> numerator:=numer(f(x));

$$numerator := 2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27$$

> denominator:=denom(f(x));

$$denominator := x^4 - 4x^3 + 5x^2 - 4x + 4$$

We form the original improper rational fraction.

> original:=numerator/denominator;

$$original := \frac{2x^5 - 8x^4 + 15x^3 - 10x^2 - 9x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}$$

Since we have an improper rational fraction and the method of partial fractions is for proper rational fractions (degree of numerator less than degree of denominator), we use `quo` and `rem` to get the quotient and remainder from a long division of polynomials. Notice that the arguments for each command are "dividend, divisor, variable."

> quotient:=quo(numerator, denominator, x);

$$quotient := 2x$$

> remainder:=rem(numerator, denominator, x);

$$remainder := 5x^3 - 2x^2 - 17x + 27$$

We form a rational fraction by dividing the remainder by the divisor.

> rf:=remainder/denominator;

$$rf := \frac{5x^3 - 2x^2 - 17x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}$$

We rewrite the original improper fraction as the sum of the quotient and a proper fraction.

> new:=quotient+rf;

$$new := 2x + \frac{5x^3 - 2x^2 - 17x + 27}{x^4 - 4x^3 + 5x^2 - 4x + 4}$$

Again, we can use the [convert](#) command to convert the proper rational expression to partial fractions.

> rf:=convert(rf,parfrac,x);

$$rf := \frac{2x + 7}{x^2 + 1} + \frac{3}{x - 2} + \frac{5}{(x - 2)^2}$$

The entire integrand is the sum of the quotient and the partial fraction decomposition of the proper fraction.

> integrand:=quotient+rf;

$$integrand := 2x + \frac{2x + 7}{x^2 + 1} + \frac{3}{x - 2} + \frac{5}{(x - 2)^2}$$

We make this into a Maple function *g* using the [unapply](#) command.

> g:=unapply(integrand,x);

$$g := x \rightarrow 2x + \frac{2x + 7}{x^2 + 1} + \frac{3}{x - 2} + \frac{5}{(x - 2)^2}$$

We integrate *g*.

> Int(g(x),x)=int(g(x),x);

$$\int \left(2x + \frac{2x + 7}{x^2 + 1} + \frac{3}{x - 2} + \frac{5}{(x - 2)^2} \right) dx = x^2 - \frac{5}{x - 2} + 3 \ln(x - 2) + \ln(x^2 + 1) + 7 \arctan(x)$$

Substitution or Change of Variable

Maple has a [Student](#) package which is designed to illustrate calculus concepts in a step by step manner. We load this package by using the [with](#) statement.

> with(Student);

[*Basics, Calculus1, LinearAlgebra, MultivariateCalculus, NumericalAnalysis, Precalculus, SetColors, SetDefault, SetDefaults, Statistics, VectorCalculus*]

We are given a list of subpackages, and we choose [Calculus1](#).

> with(Calculus1);

[*AntiderivativePlot, AntiderivativeTutor, ApproximateInt, ApproximateIntTutor, ArcLength, ArcLengthTutor, Asymptotes, Clear, CriticalPoints, CurveAnalysisTutor, DerivativePlot, DerivativeTutor, DiffTutor, ExtremePoints, FunctionAverage, FunctionAverageTutor, FunctionChart, FunctionPlot, GetMessage, GetNumProblems, GetProblem, Hint, InflectionPoints, IntTutor, Integrand, InversePlot, InverseTutor, LimitTutor, MeanValueTheorem, MeanValueTheoremTutor, NewtonQuotient, NewtonsMethod, NewtonsMethodTutor,*

PointInterpolation, RiemannSum, RollesTheorem, Roots, Rule, Show, ShowIncomplete, ShowSolution, ShowSteps, Summand, SurfaceOfRevolution, SurfaceOfRevolutionTutor, Tangent, TangentSecantTutor, TangentTutor, TaylorApproximation, TaylorApproximationTutor, Understand, Undo, VolumeOfRevolution, VolumeOfRevolutionTutor, WhatProblem]

A list is given of all the new commands added. Our interest is in [Rule](#).

We can use [Rule](#) for a change of variable. For example, suppose we wish to use the substitution $u = 2x$

in the integral $\int_a^b \frac{1}{\sqrt{1-4x^2}} dx$. It is implemented as follows.

```
> Rule[change,u=2*x](Int(1/sqrt(1-4*x^2), x = a .. b));
```

$$\int_a^b \frac{1}{\sqrt{-4x^2 + 1}} dx = \int_{2a}^{2b} \frac{1}{2\sqrt{-u^2 + 1}} du$$

We use value to integrate the right-hand-side.

```
> Int(1/sqrt(1-4*x^2), x = a .. b)=value(rhs(%));
```

$$\int_a^b \frac{1}{\sqrt{-4x^2 + 1}} dx = -\frac{1}{2} \arcsin(2a) + \frac{1}{2} \arcsin(2b)$$


A Warning

Consider the function $f(x) = \frac{1}{x}$. This function is not continuous at 0. Let's look at its graph.

```
> f:=1/x;
```

$$f := \frac{1}{x}$$

```
> plot(f,x=-5..5,y=-5..5,discont=true);
```



Now let us look at its indefinite integral.

> Int(f,x)=int(f,x);

$$\int \frac{1}{x} dx = \ln(x)$$

Where is the absolute value about x ? What is happening is that Maple assumes that **ln** is the complex valued version of the function. Let's see what happens for definite integrals, starting with positive limits of integration.

> Int(f,x=3..5)=int(f,x=3..5);

$$\int_3^5 \frac{1}{x} dx = -\ln(3) + \ln(5)$$

> evalf(rhs(%));

0.510825623

This looks good. Now let's try negative limits of integration.

> Int(f,x=-5..-3)=int(f,x=-5..-3);

$$\int_{-5}^{-3} \frac{1}{x} dx = -\ln(5) + \ln(3)$$

```
> evalf(rhs(%));
```

```
-0.510825623
```

This looks good, too. Notice how it uses the absolute values of -3 and -5 instead of -3 and -5. Now lets try a negative and a positive limit of integration.

```
> Int(f,x=-5..5)=int(f,x=-5..5);
```

$$\int_{-5}^5 \frac{1}{x} dx = \text{undefined}$$

This indicates that it can't do it, which is good.

Integral Tutor

Finally, we look at the [integral tutor](#) , which is also part of the **Calculus1** package.

```
> IntTutor(Int(exp(x)*sin(x), x));
```

$$\int e^x \sin(x) dx$$
$$\int e^x \sin(x) dx$$